



AD-A165 855 THE BASIC INSTRUCTIONAL PROGRAM STUDENT MANUAL
MAINSAIL CONVERSION (U) STANFORD UNIV CA INST FOR
MATHEMATICAL STUDIES IN THE SOCIAL S.
UNCLASSIFIED M L DAGEFORDE ET AL APR 78 NRPDC-SR-78-9 F/G 9/2

1/1

NL





MICROCOPY RESOLUTION TEST CHART
NATIONAL BUREAU OF STANDARDS 1962 A

AD-A165 855



NAVY PERSONNEL RESEARCH AND DEVELOPMENT CENTER, WASHINGTON, D.C. 20375

1
DTIC
SELECTED
MAR 24 1986
S D
D

NPRDC SR 78-9

APRIL 1978

THE BASIC INSTRUCTIONAL PROGRAM STUDENT
MANUAL: MAINSAIL CONVERSION

DTIC FILE COPY

DISTRIBUTION STATEMENT A
Approved for public release
Distribution Unlimited

86 2 28 055

NPRDC SR 78-9

April 1978

THE BASIC INSTRUCTIONAL PROGRAM
STUDENT MANUAL: MAINSAIL CONVERSION

Mary L. Dageforde
Marian Beard
Avron V. Barr

Institute for Mathematical Studies in the Social Sciences
Stanford University
Palo Alto, California 94305

Reviewed by
John D. Ford, Jr.

Navy Personnel Research and Development Center
San Diego, California 92152

FOREWORD

This research and development was conducted in response to Navy Decision Coordinating Paper, Education and Training Development (NDCP-Z0108-PN) under subproject Z0108-PN.32, Advanced Computer-Based Systems for Instructional Dialogues, and the sponsorship of the Director, Naval Education and Training (OP-99). The overall objective of the subproject is to develop and evaluate advanced techniques of individualized instruction.

This report is one in a series of six dealing with the BASIC (Beginner's All-Purpose Symbolic Instruction Code) Instructional Program (BIP), which is a "tutorial" programming laboratory designed for the student with no previous training in programming. The others concern (1) the original BIP student manual (Note 1), (2) the conversion of BIP into the MAINSAIL programming language (Note 2), (3) BIP system documentation (Note 3), (4) the BIP supervisor's manual (Note 4), and (5) curriculum information networks for computer-assisted instruction (Beard, Barr, Gould, & Wescourt, 1978). This report differs from Note 1 in that it incorporates changes resulting from the MAINSAIL conversion.

This report is intended for use by students using the BIP system. The work was performed under Contract N00123-76-C-1543 to Stanford University. The contract monitors were Dr. John D. Fletcher and Dr. James D. Hollan.

**J. J. CLARKIN
Commanding Officer**

SUMMARY

The BASIC Instructional Program (BIP) is a "hands-on laboratory" that teaches elementary programming in the BASIC language. This manual is the student's main source of information about the BIP system and the BASIC language. It is organized as a reference document aimed at students with no previous programming experience.

CONTENTS

	Page
SECTION 1. INTRODUCTION	1
1.1 The BASIC Language and the BASIC Instructional Program (BIP)	1
1.2 Using the Manual	1
1.3 Signing On	1
1.4 Talking to BIP	2
1.5 A Sample Interaction with BIP	3
1.6 Some Helpful Keys to Know	8
1.7 Error Messages and Changing Your Program	8
SECTION 2. PROGRAMMING IN BASIC WITH BIP	11
2.1 Programming	11
2.2 Program Storage and Execution	12
2.3 Line Numbers	13
2.4 END	13
2.5 Input/Output	14
2.6 PRINT	14
2.7 Data Types and Values	16
2.8 Primaries	16
2.9 BASIC Variables	18
2.10 Assignment	18
2.11 LET (Assignment)	19
2.12 Expressions and Operators	20
2.13 BASIC Operators	22
2.14 INPUT	24
2.15 READ . . . DATA and REOPEN	25
2.16 DIM	26
2.17 Program Flow	28
2.17.1 Loops	28
2.17.2 Branch and Return	30
2.18 GOTO	31
2.19 Relational and Boolean Operators	32
2.20 IF . . . THEN	34
2.21 FOR . . . NEXT	35
2.22 GOSUB . . . BEGINSUB . . . RETURN . . . ENDSUB	39
2.23 Functions, Arguments, and Returning Values	40
2.23.1 Built-in Functions	41
2.23.2 RND	41
2.23.3 INT	41
2.23.4 SQR	43
2.23.5 LEN	43
2.23.6 User-Defined Functions	44



Availability Codes	
Dist	Avail and/or Special
A-1	

	Page
2.24 Other Useful Statements	45
2.24.1 STOP	45
2.24.2 REM	45
SECTION 3. BIP COMMANDS	47
3.1 Curriculum Manipulation	47
3.2 Program Manipulation	48
3.3 File Storage and Access	51
3.4 Dealing With the World	52
GLOSSARY	53
REFERENCE	61
REFERENCE NOTES	61
DISTRIBUTION LIST	63

SECTION 1. INTRODUCTION

1.1 The BASIC Language and the BASIC Instructional Program (BIP)

This course is designed to help you learn some fundamental programming concepts through the BASIC language. BASIC is widely used; it is probably available on almost any computer system you are likely to encounter. BIP is an acronym for "BASIC Instructional Program," the program that runs this course. It is used only for this purpose and you will never hear of it in another context.

The version of BASIC used in this course is not identical with the many other versions you may find elsewhere. However, the fundamentals are the same, and the transition to another version of BASIC will be easy.

1.2 Using the Manual

This manual is meant to be an easy and fast source of reference material. It will be most effective if you have it with you while you are working at the terminal. Try to become familiar with the manual, but do not try to memorize it. Keep it handy and refer to it often.

The first section of the manual introduces you to BIP and some of the keys on the terminal that you should know about. The main body of the manual is the second section, which explains fundamental programming concepts and structures and describes the language in which you will write your own programs (namely, BASIC). The third section lists and explains BIP's special commands. The glossary lists all the specialized terms used in the manual, and refers to the appropriate sections for further information.

The manual is not intended to be a task-by-task guide to the course. It is a reference manual that contains a complete description of all the BASIC statements (the "sentences" of the language) and BIP commands. Especially when you first start programming, a reference manual contains a large amount of information that you are not ready to use. You must try to isolate exactly what you're looking for, and to ignore information that doesn't seem to relate to your immediate problem. This is not easy, but it becomes easier with practice. The glossary is usually a good place to start.

Advice: Don't be afraid to make mistakes. A computer is a consistent machine, and you can frequently discover what works and what doesn't by trying different ways of doing something and watching the results carefully. The manual is full of sample programs that illustrate how BASIC works. Copy and RUN these programs whenever you like.

1.3 Signing On

Whenever you want to use a computer, you always have to start by establishing communication with the machine somehow, letting the computer know who you are and what you want to do. Ask your supervisor how to originally sign on to the computer and start BIP running. Then the terminal will say

WELCOME TO BIP!!

Please type your number and first name.

Type your BIP number, a space, your first name, and a carriage return (which is probably a key marked "CR" or "RETURN" on your keyboard, abbreviated by <cr> throughout this manual). The terminal will say "HI"; and you are signed on. You will sign on in this way every time you work with BIP.

In case you make a typing mistake, there may be a key marked "DEL" or "DELETE" on your keyboard that erases the last character you typed, like a backspace. If not, go ahead and type a <cr>. BIP will tell you it doesn't know who you are and will sign you off automatically. Then you can start all over and do your sign on correctly.

Once you have signed on, you will be "talking" to BIP. You must type <cr> to end each line you type. BIP reads and responds to your commands after you type <cr>. BIP types a * every time it is ready for you to type something.

It is not too soon to tell you about signing off. You must sign off before you leave the terminal. Do it by typing BYE <cr> to BIP. The terminal should print a short message ending with GOODBYE.

Please do not leave a terminal that has not said GOODBYE to you.

Occasionally, you will be the victim of a "system error" or a "system crash." These are unexpected, unpredictable, unavoidable events. You will know that one has occurred either because your terminal suddenly prints something like "SORRY, SYSTEM ERROR" or because your terminal stops printing anything at all. If you are near any other people using the same computer, you can ask them whether they are still getting any response; if they are, and you aren't, you should probably find the person who knows something about BIP.

1.4 Talking to BIP

BIP does not present lessons on programming. It does not ask questions and wait for you to type correct answers. It does present programming tasks that require you to write BASIC programs. By writing, running, testing, and fixing your own programs, you will learn a lot about programming. BIP will help you, not by knowing the correct answers (many different programs can produce the "right" result) but by identifying errors, giving you more information, and presenting tasks that build on the skills you have developed.

The pattern of the interaction between you and BIP generally goes like this:

- a. You ask for a task, by typing TASK. BIP prints out the requirements for a program that it expects you to write and run.

- b. You write the program, test it, fix it, test it, and complete it. You will make a number of errors along the way, many of which will cause BIP to print an error message, telling you that it can't understand what you typed, or can't do what you told it to do.
- c. Having written the program required by the task statement, you type MORE, and BIP looks at your program to see that it works as it should. BIP then completes the task by giving you the "post-task interview." In some cases, the current task will be extended with some additional requirements.

Within that pattern, many additional things may happen. You may be confused, either by the statement of the problem, by the error messages printed by BIP, or by your program doing something you do not expect when you run it. There are specific ways to deal with each kind of confusion.

- a. If the task is not clear, you may request a HINT or use REP. To get a better idea of the output your program should produce, you may run a DEMO.
- b. If the meaning of the error messages is not clear, you may type a question mark (?) just after the message is printed. BIP will give you some brief information about that error. You should also look in the glossary for more information, starting with the first word or phrase that you don't understand. Use the cross-references to find the information you need.
- c. If your own program does not behave as you want it to, you should first LIST it to see exactly what it says. You should also use the debugging option -TRACE- that will show you, in detail, your program in action.

If you are already confused just by reading this, go to the glossary at the end of the manual and start looking up the words printed here in upper case, like TASK. Do not try to memorize anything. Do try to get an idea of how you can find information in the manual.

1.5 A Sample Interaction with BIP

This dialogue should give you an idea of what it's like to work with BIP and illustrate how you can learn from your errors. The student's typing is underlined; everything else in upper case is printed by BIP. (You will never have to type underlines yourself when you use BIP.) Comments and explanations appear at the right in lower case. Remember that each line of student's typing ends with a <cr>.

Start BIP running, and sign on.

WELCOME TO BIP !!

PLEASE TYPE YOUR NUMBER AND FIRST NAME.

123 CHARLES

HI CHARLES!

*TASK

You ask for something to do.

BEGIN TASK: STRINGY

USE THE -SCR- COMMAND TO SCRATCH (ERASE) YOUR PREVIOUS PROGRAM, IF ANY.

WRITE A PROGRAM THAT PRINTS THE STRING "SCHOOL"; START WITH THE GLOSSARY
TO FIND OUT WHAT A STRING IS.

BIP describes a program for you to write.

*SCR

First, get rid of any old program that's
left over from a previous task.

*HINT

You're not sure how to start, so you
ask for a HINT.

HINT #1

YOUR PROGRAM SHOULD HAVE TWO STATEMENTS IN IT: ONE -PRINT- STATEMENT,
AND ONE -END- STATEMENT. DON'T FORGET ABOUT LINE NUMBERS, AND DON'T
FORGET THAT STRINGS MUST HAVE QUOTE MARKS AROUND THEM.

*DEMO

You take a look at what the model does.

EXECUTION OF THE MODEL.

SCHOOL

EXECUTION COMPLETED AT LINE 99

DEMO ENDED

As you thought, the model is a program
that prints "SCHOOL" FOR YOU.

*PRINT

You type what you think is a -PRINT-
statement.

BASIC STATEMENTS MUST HAVE A LINE NUMBER
LINE NOT ACCEPTED. (TYPE ? FOR HELP).

And you are reminded that you need a
line number.

*10 PRINT

When you use the line number, BIP
doesn't complain.

*20 "SCHOOL"

Now you try to say what it is that you
want your program to print.

MISSING BASIC OPERATOR
LINE NOT ACCEPTED. (TYPE ? FOR HELP)

But that must not be the way to do it.
You ask for more help by typing a "?"

*?

CHECK TO SEE IF YOU HAVE A CORRECT BASIC STATEMENT ON THIS LINE

BIP can't tell what to do with "SCHOOL"
all by itself. You need a BASIC state-
ment that does something with the string
"SCHOOL"; you decide to consult the
manual.

*10 PRINT "SCHOOL"

The right idea: a -PRINT- statement
that prints something.

SYNTAX ERROR: UNMATCHED QUOTE MARKS
LINE NOT ACCEPTED. (TYPE ? FOR HELP)

But something is wrong, and you ask
for more help.

*?

REMEMBER THAT ALL STRINGS MUST HAVE A QUOTE AT THE BEGINNING AND END.

TYPE ? FOR MORE HELP

That makes sense.

10 PRINT "SCHOOL"

You try again on the -PRINT- statement.

WAS: 10 PRINT

BIP didn't complain. It just reminded
you that you have replaced line 10.
Things look good.

*RUN

So you try to -RUN- the program to see
if it works.

ERROR DOCTOR DIAGNOSIS: NO 'END' AT END
LOOK AT LINE NUMBER: 10
TYPE ? FOR HELP

It doesn't even run. It needs to have
an -END- statement.

*?

You ask for more help.

EVERY PROGRAM SHOULD HAVE EXACTLY ONE -END- STATEMENT--THE LAST LINE OF
THE PROGRAM

TYPE ? FOR MORE HELP

And you understand what it means.

*20 END

So you type an acceptable -END-. You think you have written a program that does what the task asked for, so you type -MORE- to go on.

*MORE

YOU DID NOT RUN THE PROGRAM.

But you didn't -RUN- the program. You haven't actually seen it print "SCHOOL" yet.

*RUN

Try again.

EXECUTION OF YOUR PROGRAM

SCHOOL

EXECUTION COMPLETED AT LINE 20

Now it looks fine. Your program ran without any errors, and it printed the string.

*WHAT

You want to make sure that you met the requirements of the task, so you ask for its text again.

YOU ARE IN TASK STRINGY.

DO YOU WANT THE TEXT PRINTED OUT? (Y OR N) Y

Yes, please.

USE THE -SCR- COMMAND TO SCRATCH (ERASE) YOUR PREVIOUS PROGRAM, IF ANY.

WRITE A PROGRAM THAT PRINTS THE STRING "SCHOOL"; START WITH THE GLOSSARY TO FIND OUT WHAT A STRING IS.

As you thought. You can always use the -WHAT- command to review the task.

*MORE

You're ready to go on.

EXECUTION OF THE MODEL

EXECUTION OF YOUR PROGRAM

LOOKS OK!!

BIP compares your program with the model, and sees that yours is fine. Now it asks you to evaluate yourself.

POST TASK INTERVIEW

HERE IS A TYPICAL SOLUTION TO THIS TASK:

```
10 PRINT "SCHOOL"  
99 END
```

DO YOU UNDERSTAND THE SOLUTION? (Y OR N) Y

You understand why the model works.

THINK ABOUT THE SKILLS USED IN THIS TASK. FOR EACH SKILL,
TYPE Y IF YOU HAVE HAD ENOUGH WORK WITH THAT SKILL.
TYPE N IF YOU THINK YOU NEED MORE WORK ON IT.

PRINT STRING LITERAL (Y OR N) N

You think you'd like to do more
with strings and quotation marks.
BIP will remember that fact; you
can expect more strings later.

TASK STRINGY COMPLETED.

The end of this task.

*TASK

You ask for another.

BEGIN TASK: PLUSFOUR

THIS PROGRAM SHOULD ASSIGN THE VALUE 6 TO THE NUMERIC VARIABLE N,
THEN PRINT THE SUM OF N AND 4.

You see some unfamiliar terms, and
realize that you have to spend at
least a little time with the manual.

*BYE

You also realize that you don't have
any more time, so you sign off.

SIGNOFF: 18-MAY-77 17:50:18

YOU HAVE COMPLETED 1 TASK(S) THIS SESSION:

STRINGY

TOTAL TIME TO DATE: .800 HOURS
TIME ON TODAY: .067 HOURS
TOTAL SESSIONS: 2
TOTAL TASKS COMPLETED: 3

COPYRIGHT (C) 1973 BY THE LELAND STANFORD JUNIOR UNIVERSITY

GOODBYE, CHARLES.

And that's all.

1.6 Some Helpful Keys to Know

<cr> Abbreviation for the carriage return key, probably marked CR or RETURN on your keyboard. Every line you type must be ended with a carriage return.

DEL (or
DELETE) Erases the last character you typed.

HOLD Stops the screen so that you can read everything before it disappears off the top.
There may be a key marked "HOLD" on your terminal.
If not, ask your supervisor what key or keys are used for this purpose. If you have a HOLD key, just hitting it once will stop the screen within a second or so.
When you want to start the screen moving, hit HOLD again. Any other character will also start the screen moving after you stop it, but that character will also print on the screen. Ignore it.

1.7 Error Messages and Changing Your Program

"Errors" were mentioned earlier. In the context of this course, an error is something that BIP knows it cannot handle correctly. For example, if you type something like "RASK" when "TASK" was the word you meant to type, BIP will give the error message ILLEGAL BIP COMMAND because it can't do anything with the incorrect word. There are three different kinds of errors that BIP detects and tells you about:

- a. "Syntax errors" are detected immediately after you complete your line. There are rules that you must follow when you give a BIP command (like the one above) or type a BASIC statement. BIP recognizes violations of those rules and complains immediately. (An error you may make frequently is to misspell a word, as in the example.)
- b. "Error Doctor errors" are detected when you tell BIP to RUN your program. A program is a list of instructions for the computer to follow; if your program is missing some essential things, the computer can't follow the instructions. BIP recognizes the absence of these essential things, and tells you what's missing.
- c. "Execution errors" are detected as your program is running. If your BASIC program turns out to be impossible to follow at some point, BIP will try to tell you what the problem is.

It is a good idea to LIST your program before you make any changes. You must make some changes if BIP prints an error message, or if the program does not produce the results you want. To make a change, either retype correctly the line with the error or use the CHANGE command (see Section 3.2). Suppose you had the line

```
50 PRINT "THE RESULT IN GALLONS IS "; X/Y
```

and you decided (or BIP forced you) to change it to

```
50 PRINT "THE RESULT IN GALLONS IS "; Y/X
```

instead. You could retype the line, changing the positions of X and Y, or you could use the command

```
CHANGE "X/Y" TO "Y/X" IN 50
```

(or, since the "TO" and the "IN" are optional,

```
CHANGE "X/Y" "Y/X" 50).
```

BIP will always tell you what the line was before the change, as a warning in case you didn't really want to change that line. (If this is the case, you must change it back again.)

If you want to delete a line completely, type the line number and the "CR" or "RETURN" key. Then LIST the program to be sure you have what you want.

SECTION 2. PROGRAMMING IN BASIC WITH BIP

This is the main body of the manual. It is organized by complexity of concepts--the most fundamental first, the more advanced later. Since programming concepts frequently overlap, however, you will have to bounce back and forth to find the information you need in a particular situation.

Do not try to memorize the information, especially the first time you read this section. You may not even want to read this entire section of the manual at one time. Subsections that should be read together, if you choose to read chunks at a time, are:

- 2.1-2.4--Some fundamentals of programming in BASIC.
- 2.5-2.11--Input, output, assignment, and variables.
- 2.12-2.13--Expressions.
- 2.14-2.16--INPUT and READ statements.
- 2.17-2.20--Sequence and control of execution.
- 2.21-2.23--FOR, GOSUB, and functions.

Read 2.24 the first time you see STOP or REM in the model solution.

2.1 Programming

A computer is not smart. It can only do what it is instructed to do, and every tiny step must be communicated in a form that the computer can understand. A program is a list of instructions to a computer.

Writing a program involves three big stages:

- a. Specify in complete detail what the program is supposed to do.
- b. Translate your statement of the problem into a language the computer understands.
- c. Check the program to be sure that it does everything you want it to do.

The difficulty of each stage relative to the others may vary, but none of the three can ever be ignored just because the programmer thinks "it's too easy." In particular, you must not neglect the first stage, the detailed description of the problem. It is often useful to write out in English exactly what you want the program to do, and in what order. You should list the steps you would have to follow to solve the problem by yourself; if you cannot do this, you will not be able to use a computer to solve the problem. For example, you can ask a friend to give you two numbers, and you can tell him the result of multiplying those numbers together. If you think about it, you can see that there are a number of steps involved:

Ask for the first number.
Hear it and remember it.
Ask for the second number.

Hear it and remember it.
Multiply and remember the result.
Tell your friend the result.

The more specific you are in describing each step of the problem, the easier it will be to complete the second stage, where you translate your English into a programming language. A computer cannot understand English, nor can it guess at your meaning if you give it an instruction that is only close to what you meant. The rules governing the syntax, or grammar, of programming languages are rigid, and you must use the correct words, the correct punctuation, etc. Just remember that your English list of steps, although essential, is not yet a computer program; you must translate each step into a series of symbolic instructions in exactly the form that the computer, through a programming language, can accept. This becomes much easier with practice, just as in any other foreign language.

The third stage in writing a program, where you check everything to be sure it all works as you want it to, is as necessary as the other two. The computer will follow exactly the instructions you give it. If these instructions do not say precisely what you meant, the program will not quite do what you want. Because programs must be so precise, it is easy to overlook small but important details, and very few programs run "correctly" the first time. No computer will make up for your negligence, so you must check the results of your program at least as carefully as you thought out the problem in the first place. This process, called "debugging," is tedious but necessary. If a program doesn't work, it's usually the programmer's fault, not the computer's.

2.2 Program Storage and Execution

In many programming languages, you first write your list of instructions, and then tell the computer to follow all the instructions in the list. Your list is sometimes called a "stored program" because the computer must store the instructions until you tell it to begin executing them. Execution is called "running" the program.

Whether the purpose of the program is to perform complicated calculations or to play a simulated card game, it must have some information on which to operate. This information is called data, and much of the data required by a program can be stored in the program itself. In BIP, the alternative to storing the data in the program is to have the user (the person who runs the program) supply some data when the program stops and asks for it.

For example, a program whose purpose is to print a 10 by 10 multiplication table should have all its information stored within it. It is not necessary to request information when the program is actually executed--the user simply tells the computer to run that particular program. In contrast, consider a program that plays a game with the user. Such a program needs to get information as it runs, since the progress of a game cannot be planned in advance. The program must stop and ask the user for information--what move he wants to make, for example. This second kind of program is called "interactive" because it requires the programmer to plan for interaction with the user of the program as it runs.

In either type of program, the data that the program deals with must be kept in the computer such that it is accessible to the program. This is done by the use of variables of different data types, which are discussed specifically in Sections 2.7 through 2.9.

A word about "the user": Programmers usually write programs for other people to use. Whether the program calculates payroll checks or plays a card game, it will be used by someone other than the person who wrote and debugged it. As you write your own programs, remember this hypothetical person called "the user." Try to make your programs understandable and complete enough so that a friend of yours could sit down and run them without any trouble.

It's also a good idea to include "remarks" inside your program, with the -REM- statement. A remark (also called a "comment") is very simple: it's just a note to yourself that explains something about the program without affecting the way the program runs at all. You will be surprised to see how soon you can forget what an "old" program (a week old, for example) is supposed to do. REMarks that are saved as part of the program itself are handy notes to remind you.

It is not hard to write a program that does the same thing over and over, never stopping. A program that never stops is in an "endless (or "infinite") loop" which you must stop or "interrupt." BIP itself helps you watch out for this. After executing a large number of statements, BIP will stop execution, tell you it thinks your program may be in an infinite loop, and ask whether or not you want to continue execution. You should probably say "no" and -LIST- your program. Then try to figure out why it may have been in an infinite loop.

The -GOTO- section (2.18) has an example of this kind of loop.

2.3 Line Numbers

Almost all implementations of BASIC require you to number each line of your program. Each line, or statement, is an instruction to BASIC, telling it to do some specific thing. When you run a BASIC program, BASIC finds and obeys the instruction with the lowest line number, then the one with the next higher number, etc. You need not type in your statements in order, because BASIC can sort them out by line number, but you must number them in the order you want BASIC to follow. A general practice is to use multiples of 10 as your line numbers so that you have plenty of numbers available if you want to insert something between two already existing lines. BIP allows you to have up to 500 lines in a single program, but most programs will be much shorter.

2.4 END

Use: To tell the computer when it has finished executing your program.

Example:

99 END.

Remarks:

Every BASIC program must have an END statement. The END statement must have the highest line number in the program.

See STOP (2.24.1).

2.5 Input/Output

This term refers to the problem of communicating with the computer-- how you tell it to do something for you, and how you make it deliver the results in a way you can understand. Most people communicate with computers through programs, so the subjects of input and output really deal with providing information to your program that makes it provide meaningful information to you.

Input is information that goes into the program. It can be stored (as part of the program itself) when the program is written (see the -READ- and -DATA- section, 2.15), or given by the user when the program is run (see the -INPUT- section, 2.14).

Output is the visible result of a program's execution. It is frequently in the form of information printed on the user's terminal (this will be the case for all the BIP programs you write), or it may be transmitted to a line-printing device, to a magnetic tape, etc. In the case of interactive programs, it is important for the programmer to remember that the output his program prints will be read by someone else, and must be reasonably understandable. A dialogue between a person and a computer is pointless if neither understands what the other says.

2.6 PRINT

Use: To get your program to type something on the terminal.

Examples:

```
40 PRINT 4
40 PRINT Y + 10
40 PRINT A$
40 PRINT "DOG"
40 PRINT 10 < 15
40 PRINT "THE VALUE OF X IS "; X; " AND X SQUARED IS "; X^2
40 PRINT      (prints a blank line.)
```

Remarks:

Use the PRINT statement whenever you want to have your program type something. Anything surrounded by quote marks is taken literally. Anything without quote marks is "evaluated"--BASIC figures out what its value is, and PRINT prints that value.

The statement

40 PRINT "X"

prints just the letter X, because of the quote marks. The statement

40 PRINT X

makes a BASIC look up the value of the variable X, then print that number. There are no quote marks, so BASIC has to evaluate X. (Read about values, variables, and evaluation in the next few sections.)

Boolean values can be printed too. The statement

40 PRINT 10 > 9

prints TRUE on the terminal, because 10 is greater than 9.

40 PRINT 10 = 100/2

prints FALSE, because 10 is not equal to 100 divided by 2.

"Fancy" PRINT statements:

Using a semicolon between two expressions allows you to print more than one expression on a single line. You may combine different types of expressions in a PRINT statement. The semicolon allows you to PRINT both literals and variables in one statement, which can make your program's output look good. For example, you could use two PRINT statements like this:

40 PRINT "Y IS"
50 PRINT Y

which would tell the user of the program the value of the variable Y, but would take two lines of output to do it. A nicer way to do it would be like this:

40 PRINT "Y IS "; Y

which would give the same information, but all on one line.

A more complicated example: Assume that the variable X has the value 4, and the variable Y has the value 5. The statement

40 PRINT "THE SUM OF YOUR NUMBERS IS "; X+Y

will cause BASIC to print

THE SUM OF YOUR NUMBERS IS 9

The statement

40 PRINT "X + Y = 15 IS "; X+Y=15

will cause BASIC to print

X + Y = 15 IS FALSE

Remember to use spaces inside your quotation marks where you need them. Some implementations of BASIC insert a space for every semicolon, but BIP's BASIC does not.

See Variables (2.8-2.11) and Expressions (2.12, 2.13, 2.19).

2.7 Data Types and Values

Most programming languages operate on three different types of information: numeric, string, and Boolean. Many languages do not allow the programmer to combine different kinds of information in a single expression, and it is essential that you understand the differences.

Numeric information is easy to understand. A number or a numeric expression is a thing that you can add, or find the square root of.

A string is a series of characters in a particular order. (A character is something a typewriter can generate, including letters, numerals, punctuation, and spaces.) You cannot add or multiply strings as you can numbers, although most languages allow you to perform some operations on strings. In the course you are taking, your name is stored in the computer as a string, which is why the terminal can type your first and last name for you when you give the -WHO- command. A string expression is a thing that has this kind of value, as opposed to a numeric value.

Boolean information is understood by the computer to be either true or false. In most programming languages, you can tell the computer to do one thing if something is true, and another thing if it is false (see the -IF-statement, section 2.20). The value of a Boolean expression is always either true or false. (The word "Boolean" comes from the name of a mathematician named Boole.)

A word about the size of numbers and the length of strings in BIP: Although you can use very large numbers (20 digits, for example), BIP is only accurate to 10 places, so very large numbers involve very large errors. Your strings can be quite long (100 characters, for example), but you only have room for about 60 characters on a line. So you should keep your numbers to a size of 10 digits or less, and your strings to 60 characters or less. Since Boolean information has no size to speak of, enough has been said.

2.8 Primaries

When your program is executed, the computer must be able to know, or to find, the value of all the pieces of information in it. As described in the previous section, these values may be numeric, string, or Boolean.

The information that your program deals with can be extremely simple, extremely complex, or anything in between. A primary is the simplest kind of information that you can talk about, because the computer must go through at most one step to find its value. Numeric and string primaries exist in almost all programming languages, either as literals (also called constants) or as variables, requiring assignment of values.

Literals are very straightforward. What you see is what you get; a literal is taken literally. A numeric literal is what you immediately recognize as a number: for example, 7 or -6.8. A string literal is enclosed in quotation marks and is something you immediately recognize as a sequence of characters (for example, "DOG" or "***!!!"). The only slightly tricky thing about string literals is that the characters may be numerals, but the value of the string is still a string, not a number: for example, "6" cannot be added or multiplied. "6"--like "A" or "XYZ"--is just something that can be printed.

The other kind of primary is the variable. Variables are used as names for values or as "boxes" to hold values. The value of a variable is either a number or a string, depending on what was assigned to the variable.

There are two kinds of variables. A simple variable is a "box" that holds one value, either one string or one number. A subscripted variable (often called an "array variable") can hold many values, in order, all under the same name.

Simple variables are like the single boxes below. The first one is a numeric variable, because the value in the box is a number. The second is a string variable, because the value in the box is a string.

N 15 D\$ "OUCH"

In this example, the value of the variable N is 15, and the value of the variable D\$ is "OUCH" (BASIC string variables always have that dollar sign. The variable D\$ is pronounced "D string" or "D dollar").

Subscripted variables are like the multiple boxes below. Each box has only one name, but (in this example) three "slots." Each slot can hold a value of its own.

(1) (2) (3)
N 8 ... 0 ... 5 D\$ (1) "OH" ... (2) "HI" ... (3) "OH"

In this example, the value of N(1) is 8, N(2) is 0, and N(3) is 5. The variable N is being used to hold a list (or "array") of numbers. The string variable D\$ is being used to hold a list or array of strings: the value of D\$(1) is "OH"; D\$(2) is "HI"; and D\$(3) is "OH" (see above). N(1) is pronounced "N sub 1" and D\$(3) is pronounced "D string sub 3."

Each of the elements in a subscripted variable can be treated as a separate variable. Its value can be changed by an assignment statement, compared to another value, printed, etc. Subscripted variables can have as many elements or "slots" as you like. See 2.16 for more information about their use.

The important thing to remember about both literals and variables is that they do not involve any operations or calculations. In the case of literals, the value is simply the literal itself--nothing is hidden. In the case of a variable, the computer can find its value immediately by looking in the "box" named by the variable, where the value is stored.

See BASIC Variables and Assignment (2.9-2.11).

2.9 BASIC Variables

Use: To name locations (or "boxes") where values are stored.

Examples:

Y
X2
B\$

Remarks:

A numeric variable names a "box" whose contents must have some numeric value (e.g., -6 or 2.5) that can be changed by arithmetic operations like addition or division. A numeric variable must be either a single letter or a single letter and a single digit. In the above examples, Y and X2 are numeric variables.

A string variable names a "box" whose contents must have some "string" value (e.g., "HORSE") that can be changed by the string operation called "concatenation." A string variable must be a letter followed by the \$ character. In the above examples, B\$ is a string variable.

See Primaries (2.8) and Assignment (2.10-2.11).

2.10 Assignment

All programming languages make extensive use of variables, the "boxes" used to hold values. A program that deals only with literals cannot be used in any kind of general way, since nothing within the program can ever change. For example, a program that adds 2 + 4 has limited use, but a program that uses variables to hold the values of two numbers, and then adds them, is obviously more useful, since that program can add any two numbers.

The mechanism by which variables are given values is called assignment. The simplest form of assignment is this:

<variable> = <literal>

For example:

X = 5

After this assignment is done, the variable X "has the value" 5. Any reference to X (like printing it, or adding 1 to it) is actually a reference to the "box" whose name is X, the box that now has 5 in it. The value of X can be changed by another assignment, after which every reference to X will be taken as a reference to that new value.

The value assigned to a variable can be given as an expression combining two or more values. Thus the value of X could be assigned as

X = 5*4

or, assuming that the variable Y had already been assigned a value of its own,

X = Y + 1

When the computer executes an assignment statement, it follows these steps:

- a. Evaluate the expression on the right side of the "=" sign.
- b. Put that value into the "box" named by the variable on the left side of the "=" sign.

Thus, the assignment X = Y + 1 means: "Find the value of Y, add 1 to it, and then assign that result as the value of X." Note that the value of Y is not changed by this assignment. Only the variable on the left side of the = sign gets a new value. (Do not confuse your right and left hands, or your variables will seem to have strange values.)

The assignment X = X + 1 means: "Find the current value of X, add 1 to it, and assign that new value to X." If X had the value 5 before the execution of the assignment statement, it would have the value 6 after the assignment.

The contents of the variable on the left side of the "=" sign are always replaced by the value of the expression on the right side. The old value of the variable (whatever value it had before the assignment statement) is lost.

2.11 LET (Assignment)

Use: To give a value to a variable.

(Note: In BIP's BASIC, you may use either the "=" sign or the "<" (a left-arrow) sign (if your keyboard has one) in assignment statements. The "<" will print as an underline or as a left-arrow on your terminal.)

Examples:

```
10 LET X = 5
10 B$   "HELLO"
10 A2 = A2 + 1
10 X$ (1) = "RAINDROP"
```

(The word LET is optional.)

Remarks:

BASIC variables are assigned values as explained above in 2.10. Note that the "=" sign does not indicate equality in this context; instead, in assignment statements, "=" and "_" mean something more like "becomes" or "has the value of."

The assignment statement in BASIC is called the LET statement, to remind you that

LET X _ 5 and X = 5

both mean "Let X have the value 5."

Remember that right and left are different, and that

M\$ = N\$

means: "Find the value of N\$, and assign that value to M\$. This LET statement will not change the value of N\$.

A statement like

100 = X or "DOG" = M\$

will cause a syntax error from BIP, because you can't assign a value to 100 or to "DOG" either. If you want the value of X to be 100—you should say X = 100. If you want the value of M\$ to be "DOG"—you should say M\$ = "DOG" to be correct.

See Data types, Primaries, and BASIC Variables (2.7-2.9). Also see DIM (2.16).

2.12 Expressions and Operators

A primary (see 2.7) can be either a variable or a literal. In either case, the computer must go through at most one step to determine the value of a primary. An operator is a symbol that tells the computer to combine or compare two primaries in some way.

Using these definitions, an expression can be defined as either a primary

Examples: "CAT"
B

or a primary followed by an operator, followed by an expression.

Examples: X + 1
W\$ & "SONG"
(6+4) * 9
((6 + Y2) - (A + B)) / X
"DOG" & (F\$ & W\$)
R\$ (1, 3) & R\$
(A >= B) OR D\$ = "DOG")

Using the term "expression" in its own definition means that an expression can be almost infinitely complex. Programming languages follow a process of evaluating each part of the expression, and then putting it all together to find the value of the expression as a whole. (Think of how you determine the meaning of a complicated phrase like "the sister of the father of my brother's sister's son's mother." A computer determines the meaning, or value, of each part of an expression in a similar way.)

More complicated expressions are evaluated from left to right and according to the following rules:

- a. Expressions within the innermost parentheses are evaluated first.
- b. Exponentiation (^) is done before any other operations.
- c. Multiplication (*) and division (/) are done next.
- d. Addition (+) and subtraction (-) are next.

This means that you may need to use parentheses to make the computer evaluate an expression correctly. In addition, you should always use spaces and parentheses to make your expressions easy for you to read. Extra spaces or extra pairs of parentheses will not cause errors.

Some examples:

5 + 3 / 2 ^ 2 is evaluated as $5 + (3 / (2^2)) = 5.75$
 $((5+3) / 2) ^ 2$ is evaluated as $(8/2)^2 = 16$

One essential thing to remember about using operators in programs is that you must be explicit. Although a normal algebraic notation like

A + 2B

is clear to you and your algebra book, it is not clear to the computer. Any time you want the program to perform multiplication, you must say so, usually with "*" (the multiplication operator). The equivalent of the above algebraic expression is

A = 2*B.

You will also quickly notice that your terminal cannot type exponents up above the base. Exponentiation is always indicated on the same line, usually with the "^" operator. (On some terminals, there is a key with an arrow that points upward. Otherwise, use 2 asterisks.) Thus, to get 17 squared, you must use

17 ^ 2 or 17 ** 2.

(Remember, spaces are optional. 17^2 is also 17 squared.)

See Operators and Operations (2.13, 2.19).

2.13 BASIC Operators

A BASIC operator can be one of many different things. The arithmetic or numeric operators are

- ^ exponentiation
- * Multiplication
- / division
- + addition
- subtraction

The arithmetic operators work in BASIC just as they do in other programming languages, as explained in 2.12.

The BASIC string operators are

- & concatenation
- (X, Y) substrings

Concatenation is used to join together two strings. For example, suppose the value of the string variable A\$ is "HELLO " (notice the space after the "O"). And suppose the value of the variable B\$ is assigned this way:

B\$ = A\$ & "THERE."

The concatenation of A\$ and "THERE." would make the value of B\$ "HELLO THERE."

Some advice about concatenating strings: If you are putting words together (as in the HELLO THERE example), don't forget about the space between the words. If you concatenate "CAR" and "WASH" this way

"CAR" & "WASH"

the result is "CARWASH"--which may be just what you want. If you say "WELCOME" & "HOME"

you get "WELCOMEHOME"--which is probably not what you want. You can say either

	"WELCOME " & "HOME"	(space after "WELCOME")
or	"WELCOME" & " HOME"	(space before "HOME")
or	"WELCOME" & " " & "HOME"	(space quoted by itself)

all of which result in "WELCOME HOME"; this concatenation

"WELCOME"&" "&"HOME"

produces the same "WELCOME HOME" result, because the space is inside the quote marks as in the other examples. A space inside quote marks is just like any other character and becomes part of the resulting string just as any letter would. Using spaces to separate different parts of your expression makes your lines easier to read, but has no effect on how the expression is evaluated.

A substring is a part of a string. In the example above, X and Y refer to the "start" and "stop" characters in the string. For example, "PURPLE" (1, 3) means the first through the third characters in the word PURPLE. The value of "PURPLE" (1, 3) is "PUR" and that of "PURPLE" (4, 5) is "PL"; the numbers can be variables, so if the value of X were 3 and the value of Y were 5, then "PURPLE" (X, Y) would be "RPL"; the string can be a variable too, so if the value of H\$ was "PHANTOM"--then H\$ (X, Y) would be the same as "PHANTOM" (3, 5) and "ANT" would be the value.

This substring "BEAN" (5,5)

would be the fifth character in the string "BEAN" if there were five characters to begin with. If you specify a nonexistent substring like this one, the result is nothing. (See 2.14 for an explanation of the "null string.")

This substring "BEAN" (3,2)

would be the third through the second character in the string "BEAN"--if BIP could count characters backwards, but it can't. An "impossible substring" like this one will cause an execution error when BIP tries to evaluate it.

BASIC cannot evaluate an expression that contains different types of values. For example, this expression has no meaning

9 + "NINE"

because 9 is a numeric primary and "NINE" is a string primary.

See Data Types and Primaries (2.7-2.8), Variables and Assignment (2.9-2.11), and Boolean Expressions (2.19).

2.14 INPUT

Use: To allow the user of the program to give a value to a variable.

Examples:

30 INPUT N	(for a number)
30 INPUT F\$	(for a string)
30 INPUT X, B\$	(for multiple input)

Remarks:

When the INPUT statement is executed, BASIC types a colon (:) and waits for the user to type something, ending with the RETURN key. Whatever the user types becomes the value of the variable in the INPUT statement.

The only limitation in the use of INPUT involves numeric variables and is imposed when someone runs the program. If a numeric variable is specified in the program, the user must type a single number, not a string or any kind of expression. Numbers like 1492 or 6.25 will be accepted, but an expression like $3*4$ will not. BIP prints an error message and lets the user try again.

This program doubles any number the user types:

```
10 PRINT "TYPE A NUMBER AND I'LL DOUBLE IT FOR YOU"
20 INPUT Y
30 Y = Y*2
40 PRINT Y
99 END
```

This program does something simple with a string typed by the user:

```
10 PRINT "TYPE A FEW WORDS AND I'LL REPEAT THEM"
20 INPUT W$
30 PRINT W$
99 END
```

Note. When typing a string in response to an INPUT the user should not type quotation marks. Also, for strings, if the user types only the "CR" or "RETURN" key, the string variable is assigned the value "". This is called the NULL string. The null string is analogous to the number 0 (zero). It is a known value, something that has meaning: It means the string version of nothing just as zero means the numeric version of nothing. Do not confuse the null string with the character " " -- which is a space.

One INPUT statement may be used to allow the user to give values to more than one variable. For example, this program accepts two numbers and adds them.

```
10 PRINT "TYPE TWO NUMBERS, ONE AT A TIME."
20 INPUT X, Y
30 PRINT "THE SUM IS "; X+Y
99 END
```

You may specify as many variables in a "multiple input" statement as you like, always separated by a comma. When BIP's BASIC executes this statement, it prints a colon for each value to be typed by the user. Other implementations of BASIC work in a different way.

See Input/Output (2.5) and Variables (2.9).

2.15 READ . . DATA and REOPEN

Use: To assign stored values to variables.

Examples:

```
10 READ X  
50 DATA 200  
  
10 READ P  
20 READ Q  
30 READ R  
200 DATA 5, 20, 50  
  
30 READ A, B$  
80 DATA 60, "DOG"  
  
60 REOPEN
```

Remarks:

Using READ and DATA combinations allows you to store values in the program and to assign those values to variables at appropriate times. The statement

```
READ X
```

causes BASIC to take a value from the DATA statement and assign that value to the variable X. For every execution of a READ statement, there must be a corresponding DATA value.

As shown in the second example above, a DATA statement may contain more than one value. BASIC keeps track of the DATA values, and after a READ is executed, BASIC moves a pointer to the next value in the DATA statement. In that second example, the variable P would get the value 5, Q would get 20, and R would get 50.

The third example shows a multiple READ statement. Execution of a multiple READ assigns values to both variables, just as if one READ immediately followed the other. In the example, execution of line 30 would result in the assignment of 60 to the variable A and the assignment of "DOG" to the variable B\$. Use multiple READ statements whenever you want to assign values to more than one variable, all at the same time.

If a READ statement is executed, and all the DATA values have been "used," an execution error message will be printed (since no value remains to be assigned). To avoid this error, use a "dummy" value at the end of the DATA list and stop READING after that value has been used. In this program, -1 is used as the "dummy" that marks the end of the list of DATA values.

```
10 PRINT "THIS PROGRAM PRINTS SQUARES"
20 READ Y
30 IF Y = -1 THEN 90
40 PRINT Y^2
50 GOTO 20
60 DATA 5, 10, 15, 20, -1
90 PRINT "FINISHED"
99 END
```

(This program contains a loop. Read about loops in 2.17.)

There are some limitations on the values you may use in a DATA statement. First, such a value must be a literal or constant--not a variable, and not an expression. The value must be a number or a string; if it is a string, it must be enclosed in quotation marks. Second, any value given in a DATA statement must be of the same type as the variable to which it will be assigned. Note that in line 80 above the numeric value 60 corresponds to the numeric variable A, and the string value "DOG" corresponds to the string variable B\$. BASIC will give an execution error if, at the time the READ is executed, the variable and the value are of different types.

You may use as many DATA statements as you like in a program. The values given in the statements will be "used" sequentially, as required by the execution of READ statements. DATA statements can appear anywhere in the program before the END, and it is a good idea to locate your DATA in a place that makes sense to you. For example, if a section of a program requires READING values from the DATA, put the DATA statements at the end of that section so that you can easily see where the DATA values will be used.

The REOPEN statement moves the "pointer" back to the first value in the DATA list. The next READ statement will then take the first DATA value in the lowest-numbered DATA statement in the program. REOPEN is useful in situations where you want to use the same DATA values, in the same order, more than once.

See Input/Output (2.5), Data Types (2.7), and Variables (2.9).

2.16 DIM

Use: To establish the size of an array (a subscripted variable).
DIM is short for DIMENSION.

Examples:

```
10 DIM L(15)
```

```
10 DIM A$ (50)
```

Remarks:

BASIC needs to know how long an array will be before you refer to any elements or "slots" in the array (for example, before you assign any values to elements of the array). The DIM statement establishes the maximum length. The DIM statement must precede (i.e., have a lower line number than) any statement that refers to an element of the array. Usually, the DIM goes at the very beginning of the program. There must be one DIM statement for every array used in the program.

Only one DIM may be executed for a given array. In the example shown below, line 20 is executed only once each time you RUN the program. BIP will stop execution and print an error message if two DIMs are executed for the same array, or if one DIM for a given array variable is executed twice. This means that you should locate all DIMs outside any loops in your program, so that BASIC executes each different DIM only once.

Suppose your DIM statement is

10 DIM X (25)

This means that you may not use more than 25 elements in the array X. Using fewer than 25 will not cause any problems.

This is a simple program using an array. It asks the user for three words, and assigns each word to an element of the array. Then it prints the words in the opposite order.

```
10 DIM L$(3)
20 PRINT "TYPE THREE WORDS, ONE AT A TIME."
30 INPUT L$(1), L$(2), L$(3)
40 PRINT "HERE'S YOUR LIST IN THE OPPOSITE ORDER."
50 PRINT L$(3)
60 PRINT L$(2)
70 PRINT L$(1)
99 END
```

The word "index" is used in connection with arrays to mean the number that specifies each element in the array. (The word "subscript" is also used.) For example, in line 50 above, the index or subscript is the number 3, and it specifies the third element in the array L. "Index" is also used in connection with loops (see 2.17) to mean the variable that counts the number of executions of the loop. This program is like the previous example, except that it allows the user to say how long his list will be, and then uses a variable as the index, both of the loop and of the array. It also uses a variable in the -DIM- statement, after that variable has been assigned by -INPUT-.

```
10 PRINT "HOW LONG IS YOUR LIST?"
20 INPUT N
30 DIM L$ (N)
40 PRINT "TYPE YOUR WORDS."
50 FOR I = 1 TO N
60 INPUT L$ (I)
70 NEXT I
80 PRINT "HERE'S YOUR LIST IN THE OPPOSITE ORDER."
90 FOR I = N TO 1 STEP -1
100 PRINT L$ (I)
110 NEXT I
999 END
```

See Primaries (2.8), FOR .. NEXT (2.21).

2.17 Program Flow

When the computer executes a stored program, it follows a predictable path through the list of instructions that is the program. In some programming languages, the order of instructions executed depends simply on the order in which the computer encounters them from the input device (e.g., card by card from a card reader or line by line from a disk file). Other languages (including BASIC, as you know) use line numbers, and the computer executes instructions in numeric order.

In either case, all languages have the ability to tell the computer to follow a different order, to go to a different place in the list of instructions and carry on from there. This is called "branching" and it can be either unconditional or conditional. Unconditional branching refers to a change in the sequence of execution that will always be carried out regardless of anything else in the program. Unconditional branching is something like telling the computer, "Don't ask any questions, just go to a different part of the program." Conditional branching asks a question first; whether or not the change in sequence is carried out depends on some condition being true. Frequently it involves looking at a certain variable, and executing the branch if the variable has a certain value. The program specifies a decision to be made by the computer.

The ability to make appropriate decisions constitutes the "smartness" of a program. Virtually no useful program runs straight through all its statements, without ever changing the order of execution.

2.17.1 Loops

A loop is a series of statements that is executed more than once. It is an extremely useful programming structure. By using a loop, you can make the computer do the same thing many times, but you give a set of instructions only once. The general form of a loop is this:

Start the loop here.
Have the program do something.
Decide if the "something" should be done again.
If so, go back up and start the loop again.
If not, continue on from here.

The "something" can be very complex. It can be most of the program; for example, a program that plays a game can start itself again depending on what information the user gives after playing once--the whole game is inside the loop.

A large category of loops follows this general pattern:

Set a "start" value.
Set an "end" value.
Set a counter equal to the start minus 1.
Increment the counter.
Do the work.
Look at the end value--if the counter is less than the end,
go back to the "increment" place and continue from there.
Otherwise (i.e., the counter is equal to the end value),
continue from here.

A "counter" is a numeric variable that you use to count something. In this case, it counts the number of times the loop has been executed--you increment the counter (add 1 to it) each time you go through the loop. The counter is also called the "index."

This pattern is used in situations where the problem can be solved by performing the same sequence of steps, perhaps with some variations, a number of times. This is "the work." The number of times "the work" is done depends on the "start" and "end" values. For example, the following is a general program (in no programming language) that counts from 1 to 5:

```
Start = 1
End = 5
Counter = start - 1
* Counter = counter + 1
Print value of counter
If counter less than end, go to *
Print goodbye
```

} These three lines
are the loop. The
work is to print the
value of the counter.

Different problems require different variations on this general pattern. For example, the "work" may involve a more complicated set of operations, or the counter may be changed by some value other than 1, or the order in which the pattern parts are executed may need to be different. Once the general pattern is understood, however, it is easier to see which details must be changed to solve a particular problem. The following is a program (in no programming language) that counts backwards from a number typed by the user. Notice the ways in which it is different from the last example.

```
Print hello user, type me a number please
Start = whatever number the user types
End = zero
Counter = start
* Print value of counter
Counter = counter minus one
If counter greater than or equal to end, go to *
Print goodbye
```

Loops do work other than counting, of course. This final example program (in no programming language) prints the user's name as many times as he or she chooses. This program doesn't need a start or end value, because it isn't counting anything, but it does need to make a comparison to decide whether or not to go through the loop again. It also needs two string variables, one to hold the user's name, and one to hold the user's answer to the yes-or-no question.

```
Print hello user, please type your name
Username = whatever string the user types
* Print shall I say your name? yes or no, please | These five
Answer = whatever the user types | lines
If answer is no, then go to goodbye line | are the
Print value of username | loop.
Go to *
Print goodbye
```

This loop uses both a conditional branch ("if the answer is no,...") and an unconditional branch ("go to *"). Sometimes it makes sense to put the conditional branch at the top of the loop (i.e., before you do "the work"), and then unconditionally go back up and start again once you have reached the bottom, as in this example.

It is not hard to write a program that makes the computer do the same thing over and over, never stopping, in which case your program is said to be in an "infinite loop." After a large number of lines have been executed, BIP will stop execution, mention that it thinks your program is in an infinite loop, and ask you whether or not it should continue execution. You should say "no" (unless you have a very long or complicated program that you think really isn't in an infinite loop), check your program carefully to see why it might be in an endless loop, change it, and then run it again. An example of a program that has an infinite loop is given in Section 2.18.

See 2.18-2.21 for the BASIC statements used to construct loops.

2.17.2 Branch and Return

Frequently, the same set of instructions is used in many different parts of a program. An efficient way to use these instructions is to set them up in one part of the program and to branch to that part from other parts. The sequence of statements that is accessed from different parts of the program is known as a subroutine.

Since a subroutine can be "called" from different places, it is important for the computer to know where to "return" to after the statements in the subroutine have been executed. Most languages have the ability to remember the place from which execution jumped to the subroutine and then to go back to that place to continue after the subroutine.

For example, consider a program that simulates a game of blackjack. It might include a subroutine that "deals the cards" by generating random numbers and translating those numbers into cards from the deck. In blackjack, the dealer deals cards in two different situations: either at the beginning of a new hand, or when one of the players asks for another card, in addition to those he holds already. So, in the blackjack program, the card-dealing subroutine would be branched to (or "called") in those different situations. What happens afterwards depends on what was happening when the dealer dealt a card. The branch-and-return capability allows the program to go back to that place after the cards have been dealt, so that play can continue appropriately. In the first case, the program would only check to see if all the cards needed to start the game had been dealt. In the second case, it would have to ask the next player if he wanted another card.

See 2.22 for the BASIC statements used to set up subroutines.

2.18 GOTO

Use: To alter the sequence of execution of the program unconditionally.

Example:

```
70 GOTO 10
```

Remarks:

BASIC executes a program in the order of the line numbers.

When you say RUN, it finds the lowest-numbered line and executes that statement. Then it finds the next higher line number and executes the statement on that line. And so it goes--it's very simple. The above example would change that order by sending BASIC back to line 10 every time line 70 was executed.

This program will repeat itself endlessly (until BIP tells the user that it may be in an infinite loop and the user tells BIP to stop execution), counting from 1 on up.

```
10 X = 1
20 PRINT X
30 X = X+1
40 GOTO 20
50 END
```

Note that once BASIC has executed the line specified in the GOTO statement, it continues execution from that point. In this example, the order of lines executed would be

```
10,  
20, 30, 40,      (here GOTO changes things)  
20, 30, 40,      (GOTO 20 again)  
20, 30, 40,      (and again)  
etc.
```

BIP helps you discover when your program is in an infinite loop by counting the number of statement executions, stopping after a large number of them, telling you it thinks your program is in an infinite loop, and asking you whether or not to continue execution.

If your GOTO statement specifies a non-existent line, BIP will print an error message before it allows you to RUN the program.

See Program Flow (2.17).

2.19 Relational and Boolean Operators

The BASIC relational operators are

```
= equal to  
<> not equal to  
< less than  
> greater than  
<= less than or equal to  
>= greater than or equal to
```

Relational operators are used to compare two values. This comparison is called a Boolean expression, and its value is always either true or false.

In numeric expressions, the relational operators work as one normally expects them to. In string expressions, relational operators compare the strings for alphabetic order. Thus:

```
6 = 6           is true  
8.7 >= 5       is true  
4 <> 8/2       is false  
"DOG" > "CAT"   is true  
"ALPHABET" < "A"  is false
```

The Boolean operators are

```
NOT  
AND  
OR
```

Boolean operators are used to combine or change Boolean expressions. Say the variable X has the value 5,
Y has the value 99, and
A\$ has the value "YES"

Now consider the following Boolean expressions:

a. NOT

The Boolean expression $X \geq 0$ is true.
The Boolean expression $\text{NOT } X \geq 0$ is false.

The expression $Y < X^5$ is false.
The expression $\text{NOT } Y < X^5$ is true.

b. AND

An expression that includes AND is true only if all its parts are true.

X <> 4 AND Y <= 100 is true.
A\$ = "NO" and Y <= 100 is false.

c. OR

An expression that includes OR is true if any or all of its parts are true.

X <> 4 OR Y > 100	is true.
A\$ = "NO" OR X < 5	is false.
Y <> 98 OR A\$ = "YES"	is true.

Unless parentheses are used, BASIC applies the Boolean operators in this order: NOT, AND, OR. Thus,

NOT A\$ = "YES" AND Y < 100

is equivalent to

(NOT A\$ = "YES") AND (Y < 100)

and the expression is false, because NOT A\$ = "YES" is false.

A\$ = "NO" OR NOT X = 6 AND Y > 50

is equivalent to

(A\$ = "NO") OR ((NOT X = 6) AND (Y > 50))
(f) ((t) (t))

and the expression is true.

If you want to force BASIC to evaluate your Boolean expressions in a different order, use parentheses as you would with numeric expressions. For example,

NOT A\$ = "YES" OR Y < 100

is equivalent to

(NOT A\$ = "YES") OR (Y < 100)
(f) (t)

and the expression is true, because Y < 100 is true.

However,

NOT (A\$ = "YES" OR Y < 100)
(t) (t)

is false, because (A\$ = "YES" OR Y < 100) is true. Parentheses can make a difference if you need to use complicated Boolean expressions.

See Data Types and Values (2.7).

2.20 IF . . THEN

Use: To modify the order of execution so that your program can do different things in different situations.

Examples:

50 IF B > 5 THEN 150

50 IF X\$ = "OXYGEN" THEN 300

50 IF A\$ = "REPEAT" AND C > 0 THEN 10

Remarks:

The IF . . THEN statement is executed in the following way:

- a. The Boolean expression following IF is evaluated as either true or false, depending on the values and the relationship within the expression.
- b. If the Boolean expression is false, the sequence of execution does not change, and the next line executed will be the line after the line containing the IF . . THEN.
- c. If the Boolean expression is true, the next line executed will be that specified by the line number after THEN. (One may say that "control is transferred" to that different point in the program, since execution will continue from that specified line, not from the line following the IF . . THEN statement.)

This short program uses an IF . . THEN to decide whether or not to start itself over:

```
10 PRINT "TYPE YOUR NAME."
20 INPUT N$
30 PRINT "HELLO, "; N$
40 PRINT "TO START OVER, TYPE 'YES'."
50 INPUT A$
60 IF A$ = "YES" THEN 10
70 PRINT "OK. GOODBYE."
99 END
```

Note that only the word YES from the user causes the program to continue execution (again) from line 10. Anything the user types that is not YES will be taken as a NO answer. This program is another example of a loop. The number of times that the loop will be executed depends entirely on what the user types when the program is run. Try this: Copy this program, then RUN it. Use TRACE or FLOW to see how things work.

See Program Flow (2.17) and Boolean Expressions (2.19).

2.21 FOR . . NEXT

Use: To have BASIC do the counting, incrementing, and checking in a loop, automatically.

Examples:

10 REM SQUARES FROM 1 to 5	See 2.24 about REM.
20 FOR N = 1 TO 5	Establish "start" and "end."
30 PRINT N	Do something.
40 PRINT N^2	Do something else.
50 NEXT N	Add 1 to N. If N is 5 or less, go to 30 again. If N is more than 5, continue to 99.
99 END	

10 REM COUNT FROM 10 TO 1	!! counts backwards because
20 FOR N = 10 TO 1 STEP -1	the step is negative.
30 PRINT N	
40 NEXT N	
99 END	

Remarks:

FOR . . NEXT loops save the programmer some work by automatically incrementing the counter and checking its value against the top value. The general form of the FOR statement is

```
FOR <index> = <start> TO <end> STEP <howmany>
```

FOR . . . NEXT Loops are executed in this way:

- a. The "index" variable is assigned the value of <start>.
- > b. The statements following the FOR statement are executed
- ^ in order.
- ^ c. When the NEXT statement is encountered,
- ^
 - (1) The value of <howmany> is added to the index.
^ If no STEP is included, 1 is added. (The value of
^ the index moves closer to <end>.)
 - (2) If the value of the index has not passed the <end>
^ value, the statements following the FOR
^ statement are executed again--the loop is repeated
^--<-<- with the new value of the index.
 - (3) If the value of the index has passed the <end>, the
loop is not repeated, and execution continues from
the statement after the NEXT statement.

The FOR statement sets up the "start" and "end" values for the loop, and marks its beginning. The NEXT statement marks the end of the loop. The value of the index variable (N in the examples above) is changed, and checked against the "TO" value, when the NEXT statement is executed. All the "work" lies between the FOR and the NEXT.

The following three programs illustrate how loops work. All three programs do the same thing: they all count by twos from two to twenty. The first program is pretty silly, since it makes the programmer do more work than is necessary:

```
10 PRINT "COUNTING BY TWOS"  
20 PRINT 2  
30 PRINT 4  
40 PRINT 6  
50 PRINT 8  
60 PRINT 10  
70 PRINT 12  
80 PRINT 14  
90 PRINT 16  
100 PRINT 18  
110 PRINT 20  
120 PRINT "WHEW"  
999 END
```

The second program is much better, since it makes the computer do more of the work:

```
10 PRINT "COUNTING BY TWOS"  
20 N = 2  
30 PRINT N  
40 N = N + 2  
50 IF N <= 20 THEN 30  
60 PRINT "FINISHED"  
99 END
```

The third program is even better, since it takes advantage of the automatic features of the FOR . . NEXT structure:

```
10 PRINT "COUNTING BY TWOS"  
20 FOR N = 2 TO 20 STEP 2  
30 PRINT N  
40 NEXT N  
50 PRINT "THAT'S ALL, FOLKS!"  
99 END
```

It is sometimes very useful to put one loop inside another; that is, to "nest" the two loops. The following program might be used by the principal of a school to add up the number of students in each grade and in the school as a whole. The "outer loop" is indexed by the variable I, and the "inner loop" is indexed by J. The extra lines on the left show you how the J-loop is nested inside the I-loop.

```
10 T = 0  
20 S = 0  
25 REM T IS FOR TOTAL IN THE SCHOOL, S IS FOR GRADE SUBTOTALS  
30 PRINT "HOW MANY GRADES DO YOU HAVE IN THIS SCHOOL?"  
40 INPUT G  
---- 50 FOR I = 1 TO G  
60 PRINT "HOW MANY CLASSROOMS DO YOU HAVE IN GRADE "; I  
70 INPUT C  
80 FOR J = 1 TO C  
90 PRINT "HOW MANY KIDS IN CLASS "; J; " IN GRADE "; I  
100 INPUT K  
110 S = S + K  
115 REM ADD THOSE KIDS TO SUBTOTAL FOR THE GRADE  
120 NEXT J  
130 PRINT "IN GRADE "; I; " YOU HAVE "; S; " STUDENTS"  
140 T = T + S  
145 REM ADD TOTAL FOR THIS GRADE INTO THE TOTAL FOR THE SCHOOL  
150 S = 0  
155 REM SET THE SUBTOTAL BACK TO ZERO, READY FOR NEXT GRADE  
---- 160 NEXT I  
170 PRINT "IN THE WHOLE SCHOOL YOU HAVE "; T; " STUDENTS"  
999 END
```

One thing to remember when you nest loops is that the inner loop(s) must be entirely contained inside the outer loop. BIP won't let you RUN the program if it has loops like this:

```
10 FOR X = 1 TO 10  
.  
.  
40 FOR Y = 10 TO 100 STEP 10  
.  
.  
70 NEXT X  
.  
.  
90 NEXT Y
```

The NEXT for the Y-loop is outside the X-loop completely, which is not allowed.

See Program Flow (2.17).

Notice these requirements of each of the four statements:

GOSUB	50 GOSUB 800	jumps into the subroutine. Line 800 must be a BEGIN SUB.
BEGIN SUB	800 BEGIN SUB "NUMERO UNO"	beginning of the subroutine. The name (whatever you like, enclosed in quotes) is optional and has no effect except to help you see what your program is doing.
RETURN	840 RETURN	jumps to the line following the GOSUB; in this case, line 60. Use as many RETURNS as you like, for conditional branching out of the subroutine.
END SUB	870 END SUB "NUMERO UNO"	marks the end of the subroutine. It causes an automatic RETURN to (in this case) line 60. The name is optional--use it to match up with the BEGIN SUB name if it helps you.

Notice that a BIP subroutine must begin with a BEGIN SUB and end with an END SUB, and that these statements must be accessed only by the GOSUB. A BIP subroutine does not require you to use a RETURN, since END SUB includes its function. In BIP, RETURN and END SUB are similar to STOP and END: you may use as many RETURNS and STOPS as you need (including none at all), but you must use one END per program and one END SUB per subroutine.

There are no jumps into a subroutine except by a GOSUB to its BEGIN SUB, and no jumps out of a subroutine except by a GOSUB (to another subroutine), a RETURN, or an END SUB. Look at these pairs of programs for illustrations of the syntax of subroutines:

*** no jumping in ***	
This example is illegal	This example is legal
10 INPUT X	10 INPUT X
20 IF X = 1 THEN 100	20 IF X <> 1 THEN 40
.	30 GOSUB 100
.	40 STOP
.	.
.	.
100 BEGIN SUB	100 BEGIN SUB
.	.
.	.

2.22 GOSUB . . BEGIN SUB . . RETURN . . END SUB

Use: To transfer execution to a subroutine, then to return back to the same place.

Remarks:

A sequence of statements that is accessed from different parts of the program is called a subroutine. BIP subroutines are somewhat different from subroutines in other implementations of BASIC. A BIP subroutine is a sequence of statements that come between a BEGIN SUB and an END SUB. The sequence is only "called" by a GOSUB. It can terminate either with a RETURN or the END SUB, both of which cause a jump back to the line after the GOSUB that called the subroutine.

Subroutines are useful in a program that uses the same sequence of statements in a number of different situations, in that they allow the programmer to write the sequence only once and yet have it accessible from many different parts of the program. When this sequence has been executed, control returns to the place from which the sequence was called. Complicated programs are also much easier to debug if they have subroutines corresponding to the different parts of the job the program is intended to do. See "Branch and Return" in Section 2.17.2.

Example:

```
. (other lines of the program)
.
50 GOSUB 800
60 PRINT "WE RETURN FROM THE SUBROUTINE."
70 GOTO 999
.
.
800 BEGIN SUB "NUMERO UNO"
810 INPUT X
820 IF X = 1 THEN 850
830 PRINT "X IS NOT 1. YOU LOSE."
840 RETURN
850 PRINT "X IS 1. YOU GET A STAR."
860 PRINT "* * * * *"
870 END SUB "NUMERO UNO"
999 END
```

When line 50 is executed, control is transferred to line 800. Execution continues with 800, 810, and 820. If X equals 1, the next lines executed are 850, 860, 870, and then back to 60. If X is not equal to 1 at line 820, the sequence is 830, 840, and then back to 60.

*** no "flow through" into the subroutine ***	
Illegal	Legal
10 GOSUB 100	10 GOSUB 100
20 PRINT "X"	20 PRINT "X"
.	.
.	.
90 PRINT "Y"	90 STOP
100 BEGINSUB	100 BEGINSUB

(The problem with the illegal example is that, after executing the PRINT statement in line 90, BASIC would reach and execute the BEGINSUB directly in the sequence of line numbers, which is illegal. A BEGINSUB may only be executed immediately after its matching GOSUB.)

*** no jumping out ***	
Illegal	Legal
10 GOSUB 100	10 GOSUB 100
20 STOP	20 STOP
.	.
.	.
100 BEGINSUB	100 BEGINSUB
110 INPUT X	110 INPUT X
120 IF X = 1 THEN 20	120 IF X = 1 THEN 140
130 PRINT "X IS NOT 1!"	130 PRINT "X IS NOT 1!"
140 ENDSUB	140 ENDSUB

*** no subroutine calling itself ***	
Illegal	There is no right way for this.
10 GOSUB 100	BASIC is not recursive (its subroutines cannot call themselves).
.	
.	
100 BEGINSUB	
110 PRINT "IN THE SUBROUTINE!"	
120 GOSUB 100	
130 ENDSUB	

See Program Flow (2.17).

2.23 Functions, Arguments, and Returning Values

Imagine this exchange. You say, "Double this number: 6" and your friend says, "Okay: 12." To double a number is to use that number in a specific way and then to give the result back. In this example, "double" is a function, the number 6 is the argument to the function, and the number 12 (the result of doubling 6) is the value returned by the function.

A function is some defined process that produces a result. It may require no arguments, like the function that picks a random number (see RND). It may require one argument, like the function that doubles a number--you can't double something without knowing what that something is. Or it may require more than one argument, like the function that finds the smaller of two numbers--you can't say something about two numbers without knowing what they both are.

A function always returns one value.

Keep the special meanings of argument and return in mind. Don't confuse them with the regular English meanings of the words.

You may think of a function as a shorthand for some series of operations. The value returned by a function is used like any other value in the programming language you are using: you may assign it to a variable, use it in a Boolean expression, print it, etc. Some examples of functions are given in the next few pages.

To generate a random number is simply to tell the computer to pick a number. One of the most interesting uses for random numbers is in programs that play games: dealing cards, choosing a number for the user to guess, or choosing a move in tic-tac-toe, for example.

2.23.1 Built-in Functions

BASIC has several built-in functions. That is, there are some operations that are so frequently used by programmers that they have been added to the commands that the interpreter understands. The exact list of these functions will vary with the implementation of BASIC, and the list is sometimes called a "library." The following functions are built into BIP's BASIC:

2.23.2 RND

Use: To generate a random number

Examples:

```
20 X = RND
```

```
20 PRINT RND*10
```

```
20 B = INT (RND * 10 + 1)
```

Remarks:

The RND function returns a random number greater than 0 and less than 1. That is, it makes the computer "pick a number" at random the way you might pick a card from a deck. RND always picks a decimal fraction between 0 and 1, so read about INT for interesting ways to generate and use random integers.

2.23.3 INT

Use: To convert a real number into an integer.

Examples:

```
30 X = INT(7.4)
```

```
30 PRINT INT (-27.98)
```

```
30 R = INT (RND * 10 + 1)
```

Remarks:

BASIC thinks of all numbers as real numbers (i.e., as numbers with decimal fractions), not as integers. There are many situations in which a program should work with only the "integer part" of a number, and the INT function does the job.

BIP's BASIC, unlike some other implementations, interprets INT to mean "return the largest integer that is not greater than the argument." This means that:

INT (7.4) = 7
INT (-7.4) = -8

because -8 is the largest integer that is not greater than -7.4.

The argument to the INT function must evaluate as a number. INT(Y*10) is legal, but INT(A\$) is not, because A\$ cannot be a number.

Some uses of INT include:

a. Generating random integers (see RND).

The RND function returns a random number between 0 and 1-- a random decimal fraction. To create an integer, you must first multiply the random number by 10 (an integer must be at least 1), and then convert it to an integer:

INT (RND*10)

will return a random integer between 0 and 9, inclusive. The value of (RND*10) will be greater than 0 and less than 10; it will range from a low of 0.01 to a high of 9.99.

INT (RND*10 + 1)

will return a random integer between 1 and 10, since the range of values (before INT is applied) is 1.01 to 10.99. This BASIC statement assigns that random value to the variable R:

R = INT (RND*10 + 1)

In general,

INT(RND * (B - (A - 1)) + A)

will return a random integer between A and B inclusive.

b. Dividing "evenly."

If a number Y divides another number X evenly, then X/Y is an integer with no decimal fraction or "remainder." The Boolean expression

X/Y = INT (X/Y)

will be true only if X is evenly divisible by Y. For example, the Boolean expression

$13/4 = \text{INT}(13/4)$

is false, because $13/4$ equals 3.25, and $\text{INT}(3.25)$ equals 3.

But

$16/8 = \text{INT}(16/8)$

is true, because $16/8$ equals 2, and $\text{INT}(2)$ equals 2.

This program uses INT to determine if the first number given is evenly divisible by the second number:

```
10 PRINT "TYPE THE DIVIDEND"
20 INPUT X
30 PRINT "TYPE THE DIVISOR"
40 INPUT Y
50 IF X/Y = INT(X/Y) THEN 80
60 PRINT "NOT EVEN! TRY AGAIN."
70 GOTO 10
80 PRINT X; " IS EVENLY DIVISIBLE BY "; Y
99 END
```

2.23.4 SQR

Use: To return the square root of a numeric expression.

Examples:

30 S = SQR(25)

30 IF SQR (X*10) > N THEN 10

30 PRINT "THE SQUARE ROOT OF B IS "; SQR(B)

Remarks:

The SQR function finds the positive square root of its argument. The only restrictions on the argument are:

- a. It must be an expression that evaluates as a number.
- b. It must be greater than or equal to zero, since negative numbers do not have real square roots.

2.23.5 LEN

Use: To return the length of a string.

Examples:

30 INPUT T\$
40 L = LEN (T\$)

30 READ C\$
40 X = LEN (C\$)

Remarks:

The LEN function counts the number of characters in its string argument. If the value of T\$ was "TOMATO"--the function would return the value 6.

2.23.6 User-Defined Functions

Use: To return the value of any expression the programmer wants to use often.

Examples:

```
30 TWICE (N) = N*2
40 IF TWICE (I) > 100 THEN 10
50 REM BACK TO 10 IF I TIMES 2 IS BIG

30 CONCAT (R$) = R$ & R$
40 INPUT D$
50 PRINT "I'LL REPEAT AFTER YOU - "; CONCAT (D$)
```

Remarks:

Most implementations of BASIC, including BIP, allow you to define your own functions. In BIP, functions may have only one argument. Both string and numeric functions may be defined. For example,

```
10 ADDER (X) = X+1
```

defines a numeric function named ADDER, whose argument is X, and whose value is X + 1.

Defining a function to do something that you have to do more than once saves you some trouble in writing your program. For example, if your program had to generate lots of random numbers (see RND and INT, above), you might define that function, then just call it each time you needed a random number. This program is a simplified illustration:

```
10 PICKME (X) = INT (RND * X + 1)
20 REM "PICKME" WILL PICK AN INTEGER BETWEEN 1 AND X
30 PRINT "HERE'S A NUMBER BETWEEN 1 AND 10:"
40 PRINT PICKME (10)
50 PRINT "AND HERE'S A NUMBER BETWEEN 1 AND 5:"
60 PRINT PICKME (5)
99 END
```

You might copy and run this program a few times to see how all these functions work together.

You may define a given function only once in a program, but you may use as many different functions as you like. The kind of expression used in a function must match the data type of the argument: If the argument is a numeric variable, the expression must be numeric, and if the argument is a string variable, the expression must evaluate as a string. The name of the function must be at least three letters long. It can be very long (20 letters), but since the purpose of functions is to save on typing, your function names should probably be less than 10 letters long. You may not use "special characters" like periods, commas, or semicolons in the function name.

2.24 Other Useful Statements

2.24.1 STOP

Use: To tell the computer that it has finished executing your program.

Example:

```
50 STOP
```

Remarks:

Every BASIC program must have an END statement. The END statement must have the highest line number in the program.

In addition, you may use as many STOP statements as you like. STOP is equivalent to END, except that STOP may have any line number. STOP statements are useful in programs that may terminate in many ways.

BIP's BASIC always prints the number of the last line executed when a program terminates. Using STOP statements can be very valuable in debugging a program that has many parts--it can help you locate problems by causing execution to terminate under certain conditions without confusing the issue by continuing execution with wrong values. Then the line number at which the program terminated can help you see what erroneous condition occurred.

See END (2.4) and GOSUB (2.22).

2.24.2 REM

Use: To write REMarks inside your program, making it easier to understand.

Examples:

```
60 REM !!! STOP LOOPING IF X IS TOO BIG.
```

```
200 REM THE FOLLOWING 5 LINES CALCULATE THE AVERAGE:
```

Remarks:

Use a REM statement whenever you like. It does not affect the execution of your program in any way, but it gives you a way to make notes about the program as you go along, inside the program itself. You may also use a REM statement with a blank line just to make a break between blocks of lines in your program.

SECTION 3. BIP COMMANDS

Whenever you deal with BASIC, you are really communicating with the computer on two levels. One level connects you with the BASIC language and the computer's ability to execute programs written in BASIC. The other level connects you with a more general operating system, which allows you some control over the world in which your own programs live. In this course, the general system is BIP, the program that runs everything you see happening at your terminal. Through BIP, you can write and execute programs in BASIC; in addition, you are presented with programming tasks and you are allowed to save and modify your programs. Some of the commands in this section are identical to those in other implementations of BASIC and some are peculiar to BIP. You will just have to learn other commands when you use other versions of BASIC.

3.1 Curriculum Manipulation

These commands deal with the programming tasks that form the instructional base of BIP.

TASK	Start a new problem. BIP will select if for you.
HINT	Print a hint. Some tasks have no hints; some have more than one. Type HINT to help you understand what the task requires.
MORE	Continue the current problem. BIP does some checking of your program before allowing you to continue.
ENOUGH	End the current task immediately. BIP does not check your program, and keeps no record of your having entered that task.
MODEL	Print out a model solution to the current task. The model solution is not necessarily the only way to write the program. BIP does not take you out of the task.
DEMO	Execute the model solution. The demo should help you write your own program by demonstrating one possible solution to the task.
DEMO TRACE	Execute the model solution and show what's happening at the same time. BIP prints the number of each line of the model solution as it is executed and prints the value of each variable each time it is assigned. Once you have run the DEMO a few times, you know what the model solution does. Then the DEMO TRACE will help you see how the model works. See TRACE in Section 3.2. If the screen is flashing by too fast, use the HOLD key. (See Section 1.6.)

3.2 Program Manipulation

These commands do not deal with the curriculum, only with the program you are currently writing and running.

LIST Print out the current program. Use this to see what your entire program looks like--it helps. You may also list just certain lines of your program by following the command LIST with either a single line number or two line numbers, separated by dashes. For example, LIST 50 would list just line 50; LIST 40-70 would list all lines with line numbers between 40 and 70, inclusive.

SCR Delete ("scratch") the current program, wiping the slate clean so you can start afresh.

RUN Execute the program--have BASIC follow your list of instructions.

SEQ <starting> <increment>
Renumber the lines of the program. <starting> is the first line you want to have "reSEQUenced," and <increment> is the distance you want to have between the lines. For example,

SEQ 100 20

will renumber the lines in your program from line 100 upward, and each new line number will be 20 more than the line number that precedes it. (The new numbers in this example, starting at 100, would be 100, 120, 140, etc.) Use SEQ when you want to reorganize your program to make more space available between the existing lines, so that you can insert new lines into the program.

SEQ also changes the line numbers specified in GOTO, IF . . THEN, and GOSUB statements so that the program executes exactly as it did before you decided to reSEQUence the line numbers.

SEQ 10 10 is the default, meaning that if you type just SEQ, it is assumed you mean SEQ 10 10.

CHANGE "<string 1>" TO "<string 2>" IN <line range>
Change part of a line or lines without typing them all over again. This command will change every occurrence of the characters in <string 1> to the characters in <string 2> in all the lines given in <line range>. The words "TO" and "IN" are optional.

<line range> can be (1) a single line number, (2) specific line numbers, separated by commas, or (3) two line numbers separated by a dash, in which case all lines whose numbers are between those two numbers are checked. If no <line range> is given, then EVERY line in your program is checked.

This command is best illustrated by examples.
Consider the line

10 PRINT "THIS IS AN EXANPLE FOR THE CHANGE COMMAND"
with "EXAMPLE" misspelled. To fix it you could either retype the whole line or give the command

CHANGE "NP" TO "MP" IN 10

(or CHANGE "NP" "MP" 10, since "TO" and "IN" are optional). Note that if you had said

CHANGE "N" TO "M" IN 10

line 10 would be changed to

10 PRIMT "THIS IS AM EXAMPLE FOR THE CHAMGE COMMAND"
which is clearly not what you'd want.

<string 1> and <string 2> do not have to be of the same length. For example, if

"THIS IS AN EASY EXAMPLE FOR THE CHANGE COMMAND"
is what you wanted your statement to be, you could give give the command

CHANGE " E" TO " EASY E" IN 10

and then later if you decided that the word "EASY" is not what you wanted, you could eliminate it with the command

CHANGE " EASY" TO "" IN 10 .

If you wanted to change the number 10 to the number 20 in lines 30, 80, and 110 of your program, you could give the command

CHANGE "10" TO "20" IN 30,80,110

You could also give the command

CHANGE "10" TO "20" IN 30-110

as long as none of the lines between 30 and 110 have occurrences of "10" that you DON'T want to change. To change "10" to "20" everywhere in your program simply type the command

CHANGE "10" TO "20"

and it would be done.

TRACE Execute the program and show what's happening at the same time. BIP prints the number of each line as it is executed, and prints the value of each variable each time it is assigned. This is an extremely valuable debugging tool. Use it on a simple program first, to see exactly what it does. Then use it any time your program does not seem to do what you intended.

TRACE <number1>

Executes the whole program. The trace will start as soon as the line numbered <number1> is executed, and the trace continues to the end of the program. Use this command if you know that the first part of your program is correct and you want to avoid taking the time to trace through things that already work.

TRACE <number1> <number2>

Executes the whole program. In addition, it TRACES execution of all lines whose numbers are between <number1> and <number2>.

For example,

TRACE 100 200

executes the entire program, and prints line numbers and variable values between lines 100 and 200 inclusive.

Example of TRACE:

For the program:
10 FOR J = 1 TO 2
20 LET X = J
30 NEXT J
40 PRINT "FINISHED!"
99 END

Typing "TRACE" will produce this output:

TRACE STARTING AT LINE 10

10: J = 1
20: X = 1
30: J = 2
20: X = 2
30: J = 3
40: FINISHED!

99:
EXECUTION COMPLETED AT LINE 99

3.3 File Storage and Access

These commands allow you to keep your programs for later use. If you do not save a program, it will disappear when you sign off. When you save a program, you must give it a name. The name can be anything you like, but it should not contain any "special characters" like periods, commas, or semicolons. Once the program has been saved, it is called a "file."

FILES

List the names and dates of all files currently saved in permanent storage. The date and time shown tell you when the file was last SAVED. The length is the number of lines in the SAVED program.

SAVE <name>

Store the current program under the <name> given. The name must not be longer than 30 characters. The program is not affected--it is simply copied to a permanent storage area.

GET <name>

Retrieve the file of the <name> given. The current program is SCRatched and replaced by the <name> file. The permanent storage of <name> is not affected. (See comments below.)

MERGE <name>

Retrieve the <name> file from storage and add it to the current program, without SCRatching the current program. BIP will print the messages DUPLICATED LINE and WAS: . . . if the MERGED file and the current program have lines with the same line number. The "new" line from the merged file will replace the "old" line that was already part of the current program. See comments below.

KILL <name>

Erase the <name> file from permanent storage. The current program is not affected.

It is a good idea to LIST your current program before you SAVE it, to verify that it is what you want. Be careful with KILL, since it is final.

Your "current program space" and "permanent storage area" are two separate things that only communicate with each other when you use these commands. Remember that SAVE and GET make copies from the current program to permanent storage and vice versa. When you GET a file, BIP copies the file from permanent storage into your current program space, and leaves the permanent file exactly as it was. If you then make some changes to the program, you must SAVE it again if you want the changes to be permanent.

For example, suppose you have SAVED a program under the name DOG, and then sign off. The next day you GET DOG and make some changes to it. If you then say SAVE CAT, your permanent storage will have both DOG (the old version) and CAT (the new one). If you say, instead, SAVE DOG, then BIP will say "OLD VERSION DELETED" and you will have only the new version, under the name DOG. The moral is: If you want to have two versions of the program, SAVE the revision with a new name. If you don't need the old version any more, SAVE the new version with the same (old) name. If you don't SAVE it at all, the new version (your current program) will disappear when you sign off, and only the old version will be in permanent storage.

3.4 Dealing With the World

WHO	Print the name and student number of the person using the terminal. Use this if someone has left the terminal without signing off. (If you sign him off, he may lose a program, so try to find him first.)
WHAT	Print the name of the current task you are in. This also allows you to have the problem text printed out for you again, without restarting the task.
WHEN	Print the date and time. Obvious use.
FIX	Leave a message for your supervisor. Use this whenever you have a problem that you think he or she should know about. Please describe the problem as thoroughly as you can. Type the <cr> key twice to end your message.
CALC	Evaluates an expression. The expression can be numeric, string, or Boolean. For example, CALC 6+4 would make BIP print 10. Or CALC "DOG" & "FOOD" would make BIP print DOGFOOD. Or CALC 5=6 would make BIP print FALSE. CALC cannot evaluate expressions containing variables.

GLOSSARY

Words in UPPER CASE are either BIP commands or BASIC statements.

argument	The value or values operated on by a function. See 2.23.
array	Also called a "subscripted variable," a variable that may have many distinct elements, each of which can be treated as a separate variable. See 2.8, 2.16.
assignment	Associating a variable name with the contents of a location. See 2.10, 2.11.
BASIC	A widely used programming language: Beginners All-purpose Symbolic Instruction Code.
BEGINSUB	The BIP BASIC statement that starts a subroutine. See 2.22.
BIP	"BASIC Instructional Program" the program that runs this course.
Boolean expressions	Expressions whose value is either TRUE or FALSE. Used in making decisions. See 2.19.
branching	Transferring control to a different part of the program rather than following the numeric sequence of line numbers. See 2.17-2.20.
BYE	The BIP command that ends your session with the computer. See 1.3, 1.5.
CALC	The BIP command that evaluates an expression. See 3.4.
CHANGE	The BIP command that makes it possible for you to change a line or group of lines in your program without typing them over. See 3.2.
character	Anything a terminal can display: letters, numbers, punctuation, or spaces. See 2.7.
concatenation	The string operation that combines two strings into one. See 2.13.
constant	Another word for "literal." See 2.8.
counter	A numeric variable used to count something: usually incremented every time some condition is satisfied. See 2.17.

data	In general, information used by program. See 2.2.
DATA	The BASIC statement that provides values to a READ statement. See 2.15.
debugging	The process of finding and correcting errors (which computer programmers call "bugs") in your program. See 1.7, 2.24.1, 3.2.
decisions	BASIC's ability to modify the order of execution of your program, depending on certain conditions. See 2.17-2.20.
DEMO	The BIP command that executes the model, to show you how one solution to the current task works. DEMO TRACE executes the model, and traces the values of all its variables at the same time. See 3.1.
DIM	The BASIC statement that specifies the maximum number of elements in an array; usually goes at the beginning of a program using arrays. See 2.8, 2.16.
END	A required BASIC statement which must be the last line in the program. It terminates execution. See 2.4.
endless loop	Another term for "infinite loop." See 2.2, 2.17.1, 2.18.
ENDSUB	The BIP BASIC statement that ends a subroutine. See 2.22.
ENOUGH	The BIP command that terminates the current task without completing it. See 3.1.
error	Something that BASIC knows it cannot handle correctly. BIP prints out an error message to tell you what it knows about the error. See 1.7.
evaluation	The process by which BASIC determines the value of an expression. See 2.7-2.8, 2.19.
execute	Make the computer do something. BASIC is said to execute the lines of a program, i.e., to follow each instruction in the program. See 2.2.
expression	Part of a BASIC statement to be evaluated: A primary or operations on primaries. See 2.12, 2.19.

FILES	The BIP command that lists the names of the files in permanent storage. See 3.3.
FIX	The BIP command that allows you to leave a message for your supervisor. See 3.4.
FOR . . NEXT	The pair of BASIC statements that sets up a machine-made loop. See 2.17, 2.21.
function	A defined process that produces a result, e.g., RND, INT, SQR, LEN. See 2.23.
GET	The BIP command that retrieves a previously SAVED program so that you can work on it again. See 3.3.
GOSUB	The BASIC statement that causes a jump to a subroutine. See 2.17, 2.22.
GOTO	The BASIC statement that allows you to alter the sequence of execution unconditionally. See 2.17, 2.18.
HINT	The BIP command that prints a hint to help you with the current task. See 3.1.
HOLD key	A key on your terminal that will stop the screen so that you can read everything before it disappears off the top. See 1.6.
IF . . THEN	The BASIC statement that allows you to alter the sequence of execution if some condition is true. See 2.17, 2.20.
increment	To add to the value of a numeric variable, frequently a variable used as a counter.
index	In an array variable, the number in parentheses that specifies each element in the list. See 2.8, 2.16.
	In a loop, the number (counter) that keeps track of the number of times the loop has been executed. See 2.17, 2.21.
infinite loop	A program is said to be in an "infinite loop" when it does the same thing over and over, never stopping. See 2.2., 2.17.1, 2.18.
input	The set of values supplied to the program; the information on which it operates. See 2.5.

INPUT	The BASIC statement that allows the user to assign a value to a variable during execution. See 2.14.
INT	The BASIC function that returns the integer part of a real number. See 2.23.
KILL	The BIP command that erases a file from permanent storage. See 3.3.
LEN	The BASIC function that returns the number of characters in a string. See 2.23.
LET	The BASIC statement that assigns a value to a variable. See 2.11.
line number	An integer that must precede each BASIC statement; statements are executed in order of increasing line numbers. See 2.3.
LIST	The BIP command that prints out your program in the order of the line numbers. See 3.2.
literal	A primary whose value is itself (as opposed to a variable). See 2.8.
location	The place in the computer's memory where a value can be stored; the place or "box" named by a variable. See 2.10.
loop	General term for a series of statements whose execution is repeated. See 2.17, 2.21.
MERGE	The BIP command that retrieves a file from permanent storage and adds it to the current program. See 3.3.
MODEL	The BIP command that prints a typical solution to the current task. See 3.1.
MORE	The BIP command that presents the next part of a task. Type it after completing a program. See 3.1.
numeric	Having to do with numbers and their values. See 2.8.
operation	The process by which two expressions are used to specify a new value:
numeric:	Addition, subtraction, multiplication, division, exponentiation.

string:	Concatenation, substring.
relational:	An operation that compares two string or numeric expressions in some way to produce a Boolean expression.
Boolean:	An operation that combines two Boolean expressions into a new Boolean expression. See 2.12, 2.19.
operator	The symbol for an operation: + - * / ^ & (start, stop) = <> < > <= >= NOT AND OR See 2.12, 2.13, 2.19.
output	The visible results of a program's execution on the terminal. See 2.5.
primary	An expression without any operation--either a literal or a variable. See 2.7-2.8.
PRINT	The BASIC statement that produces visible results by causing the terminal to type something. See 2.6.
program	A list of instructions for a computer to follow, written in a language that the computer understands. See 2.1.
READ	The BASIC statement that assigns a value to a variable; the value is stored in the program in the DATA statement. See 2.15.
REM	The BASIC statement that does nothing. It simply allows the programmer to make notes within the program. See 2.24.
REOPEN	The BASIC statement that moves the "Read-data pointer" back to the first DATA value in the program. See 2.15.
return	To determine and give back a value. All functions return a value. See 2.23.
RETURN	The BASIC statement that causes a jump back from a subroutine to the place from which the subroutine was called. See 2.22.

RND	The BASIC function that returns a random decimal fraction between 0 and 1. It requires no arguments. See 2.23.
RUN	The BIP command that tells the computer to execute your program. See 3.2.
SAVE	The BIP command that puts your current program into permanent storage for your next session. See 3.3.
SCR	The BIP command that erases your current program. See 3.2.
SEQ	The BIP command that renbers the lines in your program to give you more available space between the existing lines. See 3.2.
signing off	Ending a session on the computer. Signing off is achieved with BYE. See 1.3.
SQR	The BASIC function that returns the positive square root of its numeric argument. See 2.23.
statement	A single BASIC instruction occupying one line of the program. See 2.1-2.3.
string	A group of characters in a particular order. See 2.7-2.8.
STOP	The BASIC statement that may appear at any place in the program and terminates execution of the program. See 2.24.
subscript	a number or numeric variable in parentheses that specifies an element of an array.
subscripted	A kind of variable, one that can contain more than one value at one time. See "array." See 2.8, 2.16.
substring	A part of a string. See 2.13.
subroutine	A sequence of BASIC statements that can be accessed and executed from different places in the main program, returning back to the place from which it is called. See 2.22.
TASK	The BIP command that presents the next programming task. Type it after completing the previous task. See 3.1.

TRACE	The BIP command that both executes a program and prints out line numbers and variables as execution progresses. See 3.2.
user	In general, the person who runs a program. Frequently, also the person who wrote it.
user-defined function	A function defined in your program, which returns the value of the expression that you specify. See 2.23.
value	The result of evaluating an expression or a function. Either a number, or a string, or TRUE or FALSE. See 2.7, 2.12, 2.19.
variable	A name for a location in the computer's memory, a "box" that can hold a numeric or string value. See 2.8-2.11.
WHAT	The BIP command that tells you the name of your current task and allows you to see the problem text again. See 3.4.
WHEN	The BIP command that tells you the date and time. See 3.4.
WHO	The BIP command that tells you who is signed on at the terminal. See 3.4.

REFERENCE

Beard, M., Barr, A. V., Gould, L., & Wescourt, K. Curriculum information networks for computer-assisted instruction (NPRDC TR 78-18). San Diego: Navy Personnel Research and Development Center, April 1978.

REFERENCE NOTES

1. Beard, M. H., & Barr, A. V. The BASIC instructional program student manual (NPRDC Special Rep. 77-2). San Diego: Navy Personnel Research and Development Center, October 1976.
2. Dageforde, M. L. The BASIC instructional program: Conversion into MAINSAIL language (NPRDC Tech. Note 78-11). San Diego: Navy Personnel Research and Development Center, April 1978.
3. Dageforde, M. L. The BASIC instructional program: System documentation (NPRDC Tech. Note 78-12). San Diego: Navy Personnel Research and Development Center, April 1978.
4. Dageforde, M. L., & Beard, M. H. The BASIC instructional program: Supervisor's manual (NPRDC Tech. Note 78-10). San Diego: Navy Personnel Research and Development Center, April 1978.

DISTRIBUTION LIST

Chief of Naval Operations (OP-987P10)
Chief of Naval Operations (OP-991B)
Chief of Naval Education and Training (N-5)
Chief of Naval Education and Training (N-7)
Chief of Naval Education and Training Support (01A), Ellyson Field,
 Pensacola, Florida 32509
Chief of Naval Technical Training (Code 015)
Chief of Naval Technical Training (Code 016)
Chief of Naval Research (Code 450) (4)
Chief of Information (OI-2252)
Commanding Officer, Naval Education and Training Program Development Center
Commanding Officer, Naval Training Equipment Center
Officer in Charge, Naval Education and Training Information Systems Activity
Director, Defense Activity for Non-Traditional Education Support
Technical Library, Air Force Human Resources Laboratory (AFSC),
 Lackland Air Force Base
Technical Training Division, Air Force Human Resources Laboratory,
 Lowry Air Force Base
Flying Training Division, Air Force Human Resources Laboratory,
 Williams Air Force Base
Personnel Research Division, Air Force Human Resources Laboratory (AFSC),
 Lackland Air Force Base
Superintendent, Naval Academy
Superintendent, Naval Postgraduate School
Superintendent, U.S. Military Academy
Superintendent, U.S. Air Force Academy
Superintendent, U.S. Coast Guard Academy
National Research Council
National Science Foundation
Science and Technology Division, Library of Congress

END

DTIC

4 - 86